ПРЕДМЕТНОЕ ОБУЧЕНИЕ
ОТКРЫТОЕ ПО В ШКОЛЕ

Иванов Олег Александрович

MAXIMA В ОБУЧЕНИИ МАТЕМАТИКЕ В ШКОЛЕ. УРОК 5. МАТЕМАТИЧЕСКИЙ АНАЛИЗ С MAXIMA

Преподавание математического анализа в школе предоставляет широчайшие возможности для использования компьютеров. Одной из методических идей, представленных в этой статье, является использование компьютера с компенсирующими целями, что особенно актуально в нашей современной российской школе.

Всякий выпускник средней школы должен уметь: а) «читать график», то есть уметь сформулировать свойства функции, график которой представлен на рисунке, и, наоборот, б) изображать эскиз графика функции, заданной некоторой формулой. Однако на первых порах учащимся трудно нарисовать график данной функции, даже после того как они провели ее стандартное исследование. Поэтому первый тип задач может быть следующим.

Задача 1. *Изобразите на экране компьютера график функции* $f(x)=\dfrac{x^2-x+1}{x^2+x+1}$, *объясните его вид, сформулировав соответствующие свойства функции* $f(x)$. *Обоснуйте свойства этой функции, проведя соответствующее исследование.*

Попросим Maxima построить график. Учтите, что первая попытка может быть не совсем удачной, поскольку аргументом оператора `wxplot2d` является не только сама функция, не только промежуток оси абсцисс, на котором изображается ее график, но может быть еще и промежуток оси ординат, в котором следует его рисовать. Итак, на экране мы увидим рис. 1.

Видно, что функция возрастает, вроде бы, на отрезках [–4; –1] и [4; 1], а убывает на отрезке [–1; 1]. При этом вроде бы ее наибольшее значение равно 3, а наименьшее лежит между 0 и 1. Что пока неясно, так это поведение функции при больших (по модулю) значениях аргумента. В связи с этим изобразим ее график на большем отрезке (рис. 2).

Теперь похоже, что на каждом из промежутков (–∞; 1] и [1; +∞) функция действительно является возрастающей, при этом $f(x)\to 1$ при $x\to\pm\infty$. Чтобы точно найти промежутки возрастания и убывания данной функции, вычислим производную и определим промежутки ее знакопостоянства. Конечно, дифференцировать надо «на бумаге», но можно дополнительно (особенно на первых порах) проверить себя, попросив Maxima найти производную заданной функции.

```
(%i3) diff((x^2-x+1)/(x^2+x+1),x);
```

(%o3) $\dfrac{2x-1}{x^2+x+1}-\dfrac{(2x+1)(x^2-x+1)}{(x^2+x+1)^2}$



```
(%i1) wxplot2d((x^2-x+1)/(x^2+x+1),[x,-4,4]);
(%t1)
```




```
(%o1)
```

Рис. 1

Как вы видите, полученное выражение хорошо бы упростить, причем так как нас интересует его знак, то лучше сразу же произвести разложение на множители.

(%i4) factor(%);

(%o4) $$\frac{2(x-1)(x+1)}{(x^2+x+1)^2}$$

```
(%i2) wxplot2d((x^2-x+1)/(x^2+x+1),[x,-20,20]);
(%t2)
```




```
(%o2)
```

Рис. 2

```
(%i5) [limit((x^2-x+1)/(x^2+x+1),x,inf),
       limit((x^2-x+1)/(x^2+x+1),x,minf)];
(%o5) [1,1]
```

Рис. 3

Теперь очевидно, что так как $f'(x) \geq 0$ при $x \in (-\infty; -1] \cup [1; +\infty)$, то данная функция возрастает на каждом из указанных промежутков, а так как ее производная неположительна на $[-1; 1]$, то на этом отрезке функция является убывающей. Воспользуемся Maxima для поиска предела этой функции «на бесконечности» (рис. 3).



`inf` есть используемое Maxima обозначение для $+\infty$, а `minf` обозначает $-\infty$.

Проведите вычисление этих пределов «на бумаге» самостоятельно.

Задача 2. *Сколько решений в зависимости от* a *имеет уравнение* $ax^5 - x + 1 = 0$*?*

Уверен, что для не очень искушенных учащихся эта задача вызовет серьезные трудности, даже если они будут использовать компьютер. Дело в том, что трудно «увидеть» ее ответ, рассматривая графики $y = ax^5$ и $y = x - 1$ или графики $y = x^5$ и $y = b(x-1)$. Конечно, данное уравнение следует переписать в виде $\frac{x-1}{x^5} = a$, с тем чтобы затем строить график его левой части. Однако и здесь учащихся будет поджидать трудность, поскольку необходимо исследовать поведение функции вблизи нуля. В связи с этим обратимся к Maxima, попросив ее нарисовать график $y = \frac{x-1}{x^5}$ (рис. 4).





Рис. 4



Красивая картинка, на которой ничего не видно, не правда ли?! Дело в том, что задача для компьютера должна быть поставлена точно не только с идейной, но и с технической точки зрения. В нашем случае следует также указать и промежуток по оси ординат, в котором следует изображать график. После того, как мы укажем, что $y \in [-1; 5]$, мы увидим следующую картинку (рис. 5).

Теперь поведение функции не вызывает сомнений, так что нам осталось лишь найти значение функции в ее точке максимума. Проведите вычисления на бумаге, чтобы убедиться в справедливости ответа, полученного при помощи компьютера (рис. 6).

Обратите внимание, что для вычисления значения производной в заданной точке необходимо использовать оператор `subst`. Именно, значение есть результат исполнения оператора `subst(x=a,diff(f(x),x))`.

Задача 3. *Решите уравнения:*

а) $2^x + x = 3$;

б) $2^x + x = 4$.

а) Положим $f(x) = 2^x + x$. Поскольку функция $f(x)$, очевидно, являются возрастающей и так как $f(1) = 3$, то $x = 1$ является единственным корнем данного уравнения.

б) Как и выше, уравнение $f(x) = 4$ имеет единственное решение. Однако, $f(1) < 4$, а $f(2) > 4$, поэтому решение данного уравнения не является целым числом. Более того, оно не является и рациональным числом (см. задание для самостоятельной работы). То, что решение существует, следует из общей теоремы о



Рис. 5

```
(%i4) f(x):=(x-1)/x^5$ f(subst(solve(diff(f(x),x),x),x));
       256
(%o5) ----
      3125
(%i6) %,numer;
(%o6) 0.08192
```

Рис. 6

промежуточном значении, однако найти его можно только приближенно. Подгрузив дополнительный пакет `newton1`, можно воспользоваться оператором `newton`, находящим приближенное значение корня данного уравнения (рис. 7).

Значением оператора `newton(f(x),x,x0,eps)` является такое число x_*, что $|f(x_*)|< eps$. При этом приближение x_* к нулю функции $f(x)$ ищется по *методу Ньютона* (*методу касательных* – о нем речь пойдет далее), в котором в качестве начального приближения взято значение x_0, являющееся одним из аргументов этого оператора.

Смысл этой задачи состоит в том, чтобы учащиеся осознали, что решения, так сказать, *существенно трансцендентных уравнений* чаще всего не могут быть найдены в явном (и привычном для школьников) виде.

Продолжим обсуждение задачи и напишем процедуру для вычисления приближенного значения корня функции. Пойдем самым простым путем, используя так называемый *метод дихотомии*. Пусть данная (непрерывная) функция принимает значения разных знаков на концах отрезка $[a; b]$. Поделим отрезок пополам и выберем ту из его половин, на концах которой функция принимает значения разных знаков, и так далее. В силу первой теоремы Больцано-Коши (см. [3]), нуль функции лежит в каждом из построенных отрезков. Остановиться можно тогда, когда длина отрезка станет меньше заданного числа (рис. 8).

Обратимся к процедуре `approot` для решения уравнения $2^x + x = 4$ (рис. 9).

Как вы видите, задача успешно решена. Тем не менее написанная процедура содержит не слишком бросающуюся в глаза ошибку (см. задачи для самостоятельного решения).

Другое дело, что рассмотренный метод не слишком эффективен. Опишем уже упоминавшийся метод Ньютона приближенного вычисления корней функции, идея которого заключается в следующем. Пусть x_0 – некоторое (начальное) приближение к корню уравнения $f(x)=0$. Касательная к графику $y=f(x)$ в точке с абсциссой x_0 задается уравнением $y=f(x_0)+f'(x_0)(x-x_0)$. Эта касательная

```
(%i1) load("newton1")$
(%i2) f(x):=2^x+x-4$ newton(f(x),x,1,0.0001);
(%o3) 1.386166993120111
```

Рис. 7

```
(%i1) apprroot(f,a,b,eps):=
      block([d:min(a,b),u:max(a,b),z],do
      if u-d<eps then return((u+d)/2) else
      (z:(u+d)/2, if f(d)*f(z)<0 then
      u:z else d:z))$
```

Рис. 8

```
(%i2) f(x):=2^x+x-4$ apprroot(f,1,2,10^(-7)),numer;
(%o3) 1.386166960000992
```

Рис. 9

пересечет ось абсцисс в точке $x_1 = x_0 - \dfrac{f(x_0)}{f'(x_0)}$, которую мы примем за следующее приближение к корню уравнения. Дальнейшие приближения строятся аналогичным образом. Условия, при которых последовательность x_n, $n = 0, 1, \ldots$, заданная рекуррентным соотношением $x_{n+1} = x_n - \dfrac{f(x_n)}{f'(x_n)}$, сходится к корню уравнения $f(x) = 0$, приведены, например, в §10.2 книги [1] (рис. 10).

Изобразим график функции

$$f(x) = 2^x + x - 4$$

и две касательные к нему, первая из которых проведена к графику в точке с абсциссой $x = 2$, а вторая – в точке, абсциссой которой является пересечение первой из касательных с осью абсцисс (рис. 11).

Понятно, что было бы бессмысленно изображать еще одну касательную (проведенную в точке с абсциссой x_2).

Напишем процедуру, реализующую метод Ньютона и проверим скорость его сходимости на примере рассматриваемого уравнения (аргумент n есть заданное число итераций) (рис. 12).

Как вы видите, уже на четвертом шаге этого алгоритма мы получили приближение с точностью 10^{-8}.

Более десяти лет назад автор читал статью Б. Куцлера (B. Kutzler), которая была посвящена использованию компьютеров для решения следующей педагогической проблемы: как научить методу Гаусса решения систем линейных уравнений студентов, не имеющих устойчивого навыка вычислений с целыми числами. Моя первая реакция состояла в том, что таких студентов не должно быть в природе, а потому

```
(%i1) f(x):=2^x-4+x$ df(x):=subst(t=x,diff(f(t),t))$
(%i3) x0:2$ a:x0-f(x0)/df(x0)$
      kas0(x):=f(x0)+df(x0)*(x-x0)$
      kas1(x):=f(a)+df(a)*(x-a)$
```

Рис. 10

```
(%i7) wxplot2d([f(x),kas0(x),kas1(x)],[x,1,2]);
```

(%t7)



(%o7)

Рис. 11

```
(%i8) newt(f,x0,n):=block([z:x0],thru n do
      z:z-f(z)/subst(x=z,diff(f(x),x)),return(z))$
(%i9) fpprintprec:12$ makelist(newt(f,2,n),n,2,4),numer;
(%o10) [1.38773277175, 1.38616752759, 1.38616698007]
```

Рис. 12

автор решает задачу для пустого множества! Но времена меняются и, к сожалению, к худшему, так что можно легко представить себе и наших российских студентов, для которых, к примеру, 3 – (–1) = –4. Поскольку для решения системы линейных уравнений придется выполнить не одно, а пару десятков подобных действий, то, как нетрудно предсказать, в каком-нибудь из них учащийся ошибется и в результате получит неверный ответ. Причем, поскольку в настоящее время чаще всего проверяется только ответ, то нельзя будет понять, из-за чего получен неверный ответ: из-за непонимания метода или же из-за ошибки в одной из арифметических операций с целыми числами. Идея Б. Куцлера состояла в использовании компьютерных пакетов для проведения собственно вычислений.

Общая педагогическая проблема может быть сформулирована следующим образом: *изучению нового материала может препятствовать отсутствие навыков, которые должны были быть получены на существенно более ранних этапах обучения*. Успехи в решении задач математического анализа сильно зависят от уровня владения учащимися техникой преобразования выражений.

Пример 1.

Требуется исследовать поведение последовательности, заданной формулой

$$x_n = \frac{n}{n^2 - n + 2}, \ n = 1,\ 2,\ \ldots$$

Конечно, надо сначала просто посчитать, для чего воспользуемся компьютером (рис. 13).

Безусловно, нужно уметь сравнивать дроби, но те, кому это сделать трудно, могут воспользоваться компьютером (рис. 14).

Ясно видно, что $x_1 = x_2 > x_3 > x_4 > \ldots$. Для доказательства рассмотрим разность

$$x_{n+1} - x_n = \frac{n+1}{(n+1)^2 - (n+1) + 2} - \frac{n}{n^2 - n + 2},$$

и это, условно говоря, есть критический момент. Некоторые из учащихся не смогут найти формулу для значения члена последовательности с номером n + 1, некоторые из тех, кто это сделал, будут долго преобразовывать полученную разность дробей. При этом их затруднения не имеют отношения к изучаемой теме! Посоветуем им обратиться к компьютеру.

```
(%i1) x(n):=n/(n^2-n+2)$ makelist(x(n),n,1,6);
(%o2) [1/2, 1/2, 3/8, 2/7, 5/22, 3/16]
```

Рис. 13

```
(%i3) fpprintprec:6$ %o2,numer;
(%o4) [0.5, 0.5, 0.375, 0.2857, 0.2273, 0.1875]
```

Рис. 14

```
(%i5) factor(x(n+1)-x(n));
(%o5) -(n-1)(n+2)/((n^2-n+2)(n^2+n+2))
```

Теперь уже ясно, что $x_{n+1} - x_n < 0$ при всех n = 2, 3, ... Таким образом, $x_{n+1} < x_n$, начиная со второго члена, что и означает, что последовательность является убывающей. При этом, с точки зрения «здравого смысла», должно быть очевидно, что данная последовательность стремится к нулю. Конечно, надо уметь доказывать этот факт, но полезно убедиться, что и компьютер умеет вычислять пределы.

```
(%i6) limit(x(n),n,inf);
(%o6) 0
```

Можно предложить учащимся использовать Maxima для построения контрпримеров.

Пример 2.

Часто приходится слышать такую, с позволения сказать, аргументацию: «Конечно, данная последовательность является убывающей, поскольку ее знаменатель растет быстрее числителя». Предложите учащимся построить контрпример к этому утверждению на примере последовательностей, заданных дробно-линейными выражениями. К примеру, рис. 15.

Пример 3.

Предложим учащимся следующее задание: «Посмотрите на результаты вычисления следующих интегралов и попробуйте обобщить их» (рис. 16).

Естественный вывод состоит в том, что первообразной функции вида $p(x)e^{\alpha x}$, где $p(x)$ есть некоторый многочлен, является функция $q(x)e^{\alpha x}$, где $q(x)$ – это многочлен той же степени, что и $p(x)$. Для поиска коэффициентов многочлена $q(x)$ имеется так называемый *метод неопределенных коэффициентов*. Положим $q(x) = ax^3 + bx^2 + cx + d$ и упростим разность $(x^3 - x + 2)e^{-2x} - (q(x)e^{-2x})'$ (рис. 17).

Приравняв нулю каждый из коэффициентов стоящего в скобках многочлена, мы и найдем значения коэффициентов искомого многочлена $q(x)$ (рис. 18).

Значением оператора `coeff(f,x,n)` является коэффициент при x^n в выражении f.

В заключение хочется еще раз подчеркнуть, что если учащийся в состоянии написать программу, реализующую некоторый метод, то это говорит о том, что этот метод им в какой-то степени понят.

```
(%i1) fpprintprec:4$ makelist((n-1)/(3*n+1),n,1,6),numer;
(%o2) [0, 0.14, 0.2, 0.23, 0.25, 0.26]
```

Рис. 15

```
(%i1) f(x):=(x^2+2)*exp(-2*x)$
      ratsimp(integrate(f(x),x));
(%o2) -(2 x^2+2 x+5)%e^(-2 x)/4
(%i3) f(x):=(x^3-x+2)*exp(-2*x)$
      ratsimp(integrate(f(x),x));
(%o4) -(4 x^3+6 x^2+2 x+9)%e^(-2 x)/8
```

Рис. 16

```
(%i5) z:ratsimp((f(x)-
      diff((a*x^3+b*x^2+c*x+d)*exp(-2*x),x)));
```

(%o5) $\left((2a+1)x^3+(2b-3a)x^2+(2c-2b-1)x+2d-c+2\right)\%e^{-2x}$

Рис. 17

```
(%i6) solve(makelist(coeff(z*exp(2*x),x,i),i,0,3),
      [a,b,c,d]);
```

(%o6) $[\,[\,a=-\frac{1}{2},\ b=-\frac{3}{4},\ c=-\frac{1}{4},\ d=-\frac{9}{8}\,]\,]$

Рис. 18

Задания для самостоятельной работы

1. Докажите, что решение уравнения $2^x + x = 4$ является иррациональным.

2. Задайте функцию, эскиз графика которой имеет вид, изображенный на рис. 19.

3. Каким образом можно, используя умение Maxima строить графики, убедиться в том, верно или нет то, что неравенство $\sqrt[5]{1+6x} \le 1+x$ справедливо при всех $x \ge 0$.

4. Найдите условие на коэффициенты p и q в уравнении, при выполнении которого данное уравнение $x^3 + px + q = 0$ имеет три действительных корня.

5. Применив процедуру `apprroot` для решения уравнения $x - 4 = 0$ на отрезке [1; 7], мы получим очень странный результат (рис. 20). Объясните его и исправьте эту процедуру.

6. Исследуйте, в зависимости от a, поведение последовательности $x_n = a^{a^{\cdot^{\cdot^{a}}}}$ (здесь возведение в степень производится $n-1$ раз).

```
(%i4) f(x):=x-4$ apprroot(f,1,7,0.01),numer;
(%o5) 6.9970703125
```

Рис. 20



Рис. 19

Ответы и решения заданий для самостоятельной работы урока 4

1. Другого способа, кроме как «раскрытие модулей», нет. Так и напишем.

```
(%i1) eqtwomod(f,g,h):=block([z,s:[],t],
      z:solve(f(x)+g(x)-h(x),x),
      for a in z do(t:subst(a,x), if (f(t)>=0)and(g(t)>=0)
      then s:append(s,[t])),
      z:solve(f(x)-g(x)-h(x),x),
      for a in z do(t:subst(a,x), if (f(t)>=0)and(g(t)<0)
      then s:append(s,[t])),
      z:solve(-f(x)+g(x)-h(x),x),
      for a in z do(t:subst(a,x), if (f(t)<0)and(g(t)>=0)
      then s:append(s,[t])),
      z:solve(f(x)+g(x)+h(x),x),
      for a in z do(t:subst(a,x), if (f(t)<0)and(g(t)<0)
      then s:append(s,[t])),
      return(s))$
```

Для примера решим уравнение, в левой части которого стоит сумма $|x-1|+|x+2|$.

```
(%i2) f(x):=x-1$ g(x):=x+2$
```

Можете поэкспериментировать с различными правыми частями уравнения, мы же ограничимся одним примером.

```
(%i4) h(x):=(x^2+5)/2$ eqtwomod(f,g,h);
(%o5) [3,1,-1]
```

2. Ответ дает следующий оператор.

```
(%i1) p(x):=a*x^3+b*x^2+c*x+d$
      solve([p(0)-p(sqrt(3)/2),p(1)-p(-1/2)],
      [a,b,c,d]);
```

(%o2) $[[a=-\frac{4\,\%r2}{3}, b=0, c=\%r2, d=\%r1]]$

Полученная запись означает, что коэффициент b многочлена должен быть равен нулю, коэффициент d произволен (что очевидно), а коэффициенты a и c должны быть связаны соотношением $3a+4c=0$. Постарайтесь понять, из каких соображений были выписаны условия на искомый многочлен. Что еще необходимо доказать, чтобы получить решение задачи?

3. Не надо иметь под руками компьютер, чтобы понять, что геометрическая прогрессия с целым знаменателем, все члены которой являются семизначными числами, может состоять из четырех членов, не более, именно: a, $2a$, $4a$, $8a$. Следовательно, далее надо искать последовательности, знаменатель которых есть рациональное число. При этом, чем ближе знаменатель к 1, тем длиннее окажется последовательность. Потому положим $q=(p+1)/p$. Пусть n – число членов прогрессии. Поскольку все они по условию – целые числа, то последовательность имеет вид $ap^{n-1}, a(p+1)p^{n-2}, \ldots, a(p+1)^{n-1}$. Все члены прогрессии окажутся семизначными числами тогда и только тогда, когда $ap^{n-1} \geq 10^6$ и $a(p+1)^{n-1} < 10^7$. Ясно, что число n будет тем большим, чем меньше числа a и p. Поэтому положим сразу $a=1$. Для числа n должны выполняться неравенства

$\dfrac{6\ln 10}{\ln p} \le n-1 < \dfrac{7\ln 10}{\ln(p+1)}$. Значит, надо найти наименьшее число p, для которого в указанном интервале найдется натуральное число. Пусть Maxima теперь посчитает.

```
(%i1) makelist([ceiling(6*log(10)/log(p)),
      floor(7*log(10)/log(p+1))],p,2,5);
(%o1) [[20,14],[13,11],[10,10],[9,8]]
```

`floor(x)` есть целая часть числа, то есть наибольшее целое число, не превосходящее данного числа x, `ceiling(x)` – это наименьшее целое число, не меньшее числа x.

Из результата вычислений следует, что наибольшее число членов прогрессии равно 11, ее первый член равен $4^{10} = 2^{20} = 1\,048\,576$, а ее последний член равен $5^{10} = 9\,765\,625$.

4. Всего существует $9 \cdot 10^{n-1}$ n-значных чисел. Если в десятичной записи n-значного числа нет единиц, то его старшей цифрой может быть 2, 3, ..., 9 – всего восемь вариантов, на остальных позициях могут находиться цифры 0, 2, 3, ..., 9 – девять вариантов, поэтому всего имеется $8 \cdot 9^{n-1}$ таких чисел. Значит, чисел, в десятичной записи которых есть хотя бы одна единица, всего имеется $9 \cdot 10^{n-1} - 8 \cdot 9^{n-1}$. Так, какое из чисел больше: $8 \cdot 9^{n-1}$ или $9 \cdot 10^{n-1} - 8 \cdot 9^{n-1}$, то есть что больше: $16 \cdot 9^{n-1}$ или $9 \cdot 10^{n-1}$? Неравенство $16 \cdot 9^{n-1} < 9 \cdot 10^{n-1}$ равносильно неравенству $0{,}9^{n-1} < 9/16$. Поскольку $0{,}9 < 1$, то последнее неравенство будет верно для всех n, начиная с некоторого значения. Таким образом, если число знаков числа достаточно велико, то чисел, в записи которых есть единица, больше, чем чисел, в записи которых ее нет. Для того чтобы определить, как много знаков должно быть, используем компьютер.

```
(%i1) for i:1 thru 20 do if 16*9^(i-1)<9*10^(i-1)
      then return(i);
(%o1) 7
```

Таким образом, если $n \ge 7$, то n-значных чисел, в записи которых есть единица, больше, чем чисел, в записи которых ее нет. Полученный результат немного противоречит нашему опыту, что связано с тем, что «по жизни» мы имеем дело с числами, количество знаков в которых не очень велико.

5. При решении уравнения с параметром ответ часто бывает достаточно длинным. То, что и в этом случае будет так, следует из следующей картинки, на которой изображены: график $y = 1 - |x-3|$ и прямые $y = x/3$, $y = x$ и $y = -x$.



Запишем данное уравнение в виде $ax = 1-|x-3|$. Ясно, что расположение точек пересечения прямой $y = ax$ с графиком $y = 1-|x-3|$ зависит от того, в каком из секторов (на которые разбили плоскость прямые $y = x/3$, $y = x$, ось ординат, и $y = -x$) находится эта прямая. Следовательно, нам стоит по отдельности рассматривать случаи: $a \ge 1$, $1/3 < a < 1$, $a = 1/3$, $-1 < a < 1/3$ и $a \le -1$. Во втором из них уравнение, очевидно, решений не имеет, в четвертом случае оно имеет два решения.

Может быть, кому-то будет удобнее переписать данное уравнение в виде $\frac{1-|x-3|}{x} = a$, с тем чтобы посмотреть на график его левой части (на рисунке ниже).



Для полноты решения приведем окончательный ответ. При $a \ge 1$ $x = -\frac{2}{a-1}$, при $1/3 < a < 1$ решений нет, при $a = 1/3$ $x = 3$, при $-1 < a < 1/3$ $x = -\frac{2}{a-1}$; $\frac{4}{a+1}$, при $a \le -1$ $x = -\frac{2}{a-1}$.

6. См.:

```
(%i1) factor((x-y)^5+(y-z)^5+(z-x)^5);
(%o1) 5(y-x)(z-x)(z-y)(z^2-yz-xz+y^2-xy+x^2)
```

Попробуйте найти обобщения разложений, найденных в этой задаче и задаче 7 урока 4.

7. Как известно, если число k/n является корнем многочлена $a_n x^n + a_{n-1}x^{n-1} + \ldots + a_0$, то k есть делитель свободного члена a_0, а n – делитель старшего коэффициента a_n этого многочлена. Поэтому все, что надо сделать, так это составить списки $k_1, k_2, \ldots$ и $n_1, n_2, \ldots$ делителей этих коэффициентов, чтобы затем проверять, является ли число $\pm k_i / n_j$ корнем данного многочлена. Встроенная помощь скажет нам, что оператор `divisors(n)` формирует список, содержащий все натуральные делители целого числа *n*. Многочлен будем задавать списком $[a_n, a_{n-1}, \ldots, a_0]$ его коэффициентов (которые удобно расположить именно в таком порядке, начиная с коэффициента при старшей степени). Для построения многочлена по набору его коэффициентов можно использовать введенную на уроке 3 процедуру `fromdig`.

```
(%i1) fromdig(a,x):=block([z:0],for t in a do z:z*x+t,
      return(z))$
```

Следующая процедура находит все рациональные корни.

```
(%i2) ratroots(a):=block([z:[],f,an:first(a),a0:last(a),
      p,q,k,n],p:divisors(a0),q:divisors(an),
      f(x):=expand(fromdig(a,x)),for n in q do
      (for k in p do(if f(k/n)=0 then z:append(z,[k/n]),
      if f(-k/n)=0 then z:append(z,[-k/n]))),return(z))$
```

Применим ее для вычисления рациональных корней многочлена $6x^3 - x^2 - 4x - 1$.

```
(%i3) ratroots([6,-1,-4,-1]);
```

(%o3) $[1, -\frac{1}{2}, -\frac{1}{3}]$

Как вы видите, она прекрасно работает.

8. Положим $p_k = k^3 + (k+1)^3 + (k+2)^3$.

```
(%i1) factor((k+2)^3-(k-1)^3);
```

(%o1) $9\left(k^2 + k + 1\right)$

Так как $p_k - p_{k-1} = (k+2)^3 - (k-1)^3$, то из проведенного Maxima вычисления следует, что числа p_k и p_{k-1} имеют одинаковые остатки при делении на 9. Поскольку p_0= 9, то методом математической индукции доказывается, что каждое из чисел p_k делится на 9.

Литература

1. *Иванов О.А.* Элементарная математика для школьников, студентов и преподавателей. М.: МЦНМО, 2009. 384 с.

2. *Иванов О.А.* Задачи по алгебре и началам анализа. СПб: БХВ-Петербург, 2005. 384 с.

3. *Пратусевич М.Я., Столбов К.М., Головин А.Н.* Алгебра и начала математического анализа. 11 класс. Учебник для общеобразовательных учреждений. Профильный уровень. М.: Просвещение, 2010. 463 с.

Иванов Олег Александрович, профессор, доктор педагогических наук и кандидат физико-математических наук, профессор кафедры общей математики и информатики СПбГУ.